
1. Overview

This document lists changes needed to correct problems in the TTCN implementation of test case 12.13 which is part of the ‘IMS_34229_IWD_13wk24’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk24’.
The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
12.13
Test Group:
IMS Call Control
ATS Version:
iwd-EUTRA-B2012-03_D13wk24
System Simulator used:
Rohde & Schwarz Multi-RAT Protocol Tester
UE used:
Intel XMM7160
Verification Status:
PASS
4. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk24’ release.
Change 1 – f_TC_12_13_IMS
	Function name
	 f_TC_12_13_IMS

	Reason for change
	1. According to 24.229, 5.1.3.1, a UE using the precondition mechanism shall indicate reliable provisional responses and preconditions. Hence, in MT cases the SS standing in for a remote UE needs to transport both 100rel and precondition in its Supported header. A prose CR has been submitted to RAN5#60 to fix this.
2. TC needs to give time to the UE/user to actually answer the call after all the setup steps rather than to immediately send BYE
3. The to-tag to be sent in BYE is first seen in the 183 Session Progress response of the UE – it cannot be concluded from the INVITE request 

	Summary of change
	1. Use send template with both precondition and 100rel. Alternatively, we might introduce a new template cs_Supported_precondition_100rel or similar
2. Introduce a moderate delay for the UE to answer the call
3. Feed the 183 Session Progress response into function to release the call. Note that we chose to implement a new function rather than changing an existing one that is used in other places as well. It is TBD if these other places are affected by this in the same manner.

	TTCN module
	 IMS_CC/12/IMS_CC_CallControlTestcases

	MCC160 Comment
	


Before change

  // See 34.229-1 clause 12.13

  function f_TC_12_13_IMS() runs on IMS_PTC

  {

    var IMS_Response_Type v_SessionInProgress183;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) INVITE_Request v_InviteRequest;

    var template (value) SDP_Message v_Tx_SDP;

    var SipUrl v_ContactUrl;

    var boolean v_IsGiba := (pc_IMS_GIBA_Sec and not pc_IMS_Sec);

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED, v_IsGiba);

    f_IMS_TestBody_Set(true);

    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);

    // Steps 1 to 15 as defined in Annex C.11

    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    // Determine SS server secure port number

    v_Tx_SDP := f_IMS_BuildSDP_AnnexC11_Step1or7("step 1");

    //  Prepare INVITE message to be sent in step 1.

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderTX(cs_Supported_precondition, v_IsGiba);
    v_InviteRequest := cs_INVITE_Request(v_ContactUrl, v_MessageHeader_Invite, cs_MessageBody_SDP(v_Tx_SDP));

    v_SessionInProgress183 := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequest, v_IsGiba);

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionInProgress183.msgHeader);

    // Establish Dedicated EPS bearer

    // See TS. 36.508 EUTRA/EPS signalling for IMS MT speech call Table 4.5A.7.3-1 Step 9-12

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    f_IMS_MTCallSetup_AnnexC11_Steps5_13(v_InviteRequest, v_SessionInProgress183);

    // MT Call is now established

    // Step 14-15. SS sends BYE to release the call; UE sends 200 OK for the BYE request and ends the call

    f_IMS_CallReleaseMT(v_InviteRequest, v_ContactUrl); 

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MT_SpeechCall);

  }
After change

  // See 34.229-1 clause 12.13

  function f_TC_12_13_IMS() runs on IMS_PTC

  {

    var IMS_Response_Type v_SessionInProgress183;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) INVITE_Request v_InviteRequest;

    var template (value) SDP_Message v_Tx_SDP;

    var SipUrl v_ContactUrl;

    var boolean v_IsGiba := (pc_IMS_GIBA_Sec and not pc_IMS_Sec);

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED, v_IsGiba);

    f_IMS_TestBody_Set(true);

    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);

    // Steps 1 to 15 as defined in Annex C.11

    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    // Determine SS server secure port number

    v_Tx_SDP := f_IMS_BuildSDP_AnnexC11_Step1or7("step 1");

    //  Prepare INVITE message to be sent in step 1.

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderTX(cs_Supported({"100rel", "precondition"}), v_IsGiba);
    v_SessionInProgress183 := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequest, v_IsGiba);

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionInProgress183.msgHeader);

    // Establish Dedicated EPS bearer

    // See TS. 36.508 EUTRA/EPS signalling for IMS MT speech call Table 4.5A.7.3-1 Step 9-12

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    f_IMS_MTCallSetup_AnnexC11_Steps5_13(v_InviteRequest, v_SessionInProgress183);

    // MT Call is now established

    f_Delay(3.0);  
    // Step 14-15. SS sends BYE to release the call; UE sends 200 OK for the BYE request and ends the call
    f_IMS_CallReleaseMT2(v_InviteRequest, v_ContactUrl, v_SessionInProgress183); 

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MT_SpeechCall);

  }
Change 2 – f_IMS_CallReleaseMT
	Function name
	 f_IMS_CallReleaseMT

	Reason for change
	1. The response carrying proper to-tag as input argument needs to be transported on
2. The response carrying proper to-tag needs to be transported to nested function constructing the BYE

3. The 200 OK for BYE is received, and not sent

	Summary of change
	1. Take previous response as input argument. Note that the existing function f_IMS_CallReleaseMT was left in place until agreement about this change is reached. So, a new function f_IMS_CallReleaseMT is defined here
Note that we introduce a new function …MT2 here instead of using the existing function …MT as the existing function is used elsewhere. However, for the other users it remains to be investigated if they would need this fix as well. For the TC under investigation the proposed fix is good enough though
2. Introduce a new function f_IMS_ByeRequest_MessageHeaderTX2 to take the response carrying the proper to-tag
3. Use receive function for the 200 OK

	TTCN module
	 Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	


Before change

  function f_IMS_CallReleaseMT(template (value) INVITE_Request p_InviteRequest,

                               SipUrl p_ContactSipUrl) runs on IMS_PTC

  {

    var template (value) MessageHeader v_MessageHeader_Bye;

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    // The SS sends BYE to release the call

    v_MessageHeader_Bye := f_IMS_ByeRequest_MessageHeaderTX(p_InviteRequest);
    IMS_Client.send(cas_IMS_BYE_Request(cs_IMS_RoutingInfo(v_Protocol), cs_BYE_Request(p_ContactSipUrl, v_MessageHeader_Bye)));

    //  The UE sends 200 OK for the BYE request and ends the call.

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderTX(v_MessageHeader_Bye))));

  }
After change

  function f_IMS_CallReleaseMT2(template (value) INVITE_Request p_InviteRequest,

                               SipUrl p_ContactSipUrl,

                               IMS_Response_Type p_SessionInProgress183 
                              ) runs on IMS_PTC

  {

    var template (value) MessageHeader v_MessageHeader_Bye;

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    // The SS sends BYE to release the call

    v_MessageHeader_Bye := f_IMS_ByeRequest_MessageHeaderTX2(p_InviteRequest, false, p_SessionInProgress183);
    IMS_Client.send(cas_IMS_BYE_Request(cs_IMS_RoutingInfo(v_Protocol), cs_BYE_Request(p_ContactSipUrl, v_MessageHeader_Bye)));

    //  The UE sends 200 OK for the BYE request and ends the call.

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Bye)))); //UE2_TC#12_13, R&S code: use receive template

  }
Change 3 – f_IMS_ByeRequest_MessageHeaderTX
	Function name
	 f_IMS_ByeRequest_MessageHeaderTX

	Reason for change
	1. The to-tag must be taken from a previous response
2. A BYE request defines a new transaction and hence needs its own branch ID
3. Max-forwards is not requested by 34.229-1
4. For To-tag see above. From-tag can be taken from either INVITE or a previous response

	Summary of change
	1. For the time being, construct new function f_IMS_ByeRequest_MessageHeaderTX2 and give it a new argument taking a previous response that defined the to-tag 
For the other users of f_IMS_ByeRequest_MessageHeaderTX it remains to be investigated if they would need this fix as well. For the TC under investigation the proposed fix is good enough though
2. Do not simply take the via header from INVITE but define it from scratch
3. Delete Max-forwards

4. Copy To header from previous response

	TTCN module
	 Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	


Before change

function f_IMS_ByeRequest_MessageHeaderTX(template (value) INVITE_Request p_InviteRequest,

                                            boolean p_MOCall := true) runs on IMS_PTC return template (value) MessageHeader
{ /* BYE (see TS 34.229, A.2.8) */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Bye := cs_MessageHeader_Dummy;

    var template (value) Via v_Via := fl_Via_AddSigCompParam(v_MessageHeader_Invite.via);
    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMT);

    var template (value) SipUrl v_InviteFromAddrSpec := f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.fromField.addressField);

    var template (value) SipUrl v_InviteToAddrSpec   := f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.toField.addressField);

    var charstring v_InviteFromTag;

    v_MessageHeader_Bye.via             := v_Via;

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Bye.cSeq            := cs_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cs_MaxForwardsDef;
    if (p_MOCall) {

      v_InviteFromTag := f_MessageHeader_GetFromTag(v_MessageHeader_Invite);

      v_MessageHeader_Bye.toField       := cs_ToWithTag(v_InviteFromAddrSpec, v_InviteFromTag);    // !!!! CHECK !!!!
      v_MessageHeader_Bye.fromField     := cs_From(v_InviteToAddrSpec, tsc_IMS_InviteToTag);       // !!!! CHECK !!!!
    } else { // MT call

      v_MessageHeader_Bye.toField       := cs_ToWithTag(v_InviteToAddrSpec, tsc_IMS_InviteToTag);  // !!!! CHECK !!!!
      v_MessageHeader_Bye.fromField     := v_MessageHeader_Invite.fromField;                       // !!!! CHECK !!!!

    }

    return v_MessageHeader_Bye;

  }
After change

  function f_IMS_ByeRequest_MessageHeaderTX2(template (value) INVITE_Request p_InviteRequest,

                                            boolean p_MOCall := true,

                                            IMS_Response_Type p_SessionInProgress183 //UE2_TC#12_13, R&S code: use previous response to get at the to-tag

                                           ) runs on IMS_PTC return template (value) MessageHeader

  { /* BYE (see TS 34.229, A.2.8) */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Bye := cs_MessageHeader_Dummy;

    var template (value) Via v_Via;                                                        

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMT);

    var template (value) SipUrl v_InviteFromAddrSpec := f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.fromField.addressField);

    var template (value) SipUrl v_InviteToAddrSpec   := f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.toField.addressField);

    var charstring v_InviteFromTag;

    // BYE needs its own branch-ID

    //So, the next lines are adapted from f_IMS_AckRequest_MessageHeaderTX()

    var charstring v_Host := f_IMS_PTC_Pcscf_Get();

    var charstring v_TransportProtocol := "UDP";

    var integer v_CallerPort := 6543;

    var template (value) ViaBody_List v_ViaBodyList;

    var charstring v_CallerAddr := tsc_IMS_CallerDomain;

    var boolean p_IsGiba := false;

    var integer v_Port_ps := f_IMS_PTC_GetPort_ps(not p_IsGiba);

    var template (omit) GenericParam v_SigCompParam := f_IMS_PTC_ImsInfo_GetSigCompParam();  // omit when sigcomp is not started

    v_ViaBodyList := {

      cs_ViaBody(v_TransportProtocol, cs_HostPort(v_Host, v_Port_ps),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch(), v_SigCompParam)),

      cs_ViaBody(v_TransportProtocol, cs_HostPort("scscf1.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol, cs_HostPort("scscf2.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol, cs_HostPort("pcscf2.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol, cs_HostPort(v_CallerAddr, v_CallerPort), f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch()))

    };  

    v_Via := cs_Via(v_ViaBodyList);  

    v_MessageHeader_Bye.via             := v_Via;

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Bye.cSeq            := cs_CseqDef(v_CSeqValue, "BYE");

    // v_MessageHeader_Bye.maxForwards     := cs_MaxForwardsDef; --commented out
    if (p_MOCall) {

      v_InviteFromTag := f_MessageHeader_GetFromTag(v_MessageHeader_Invite);

      v_MessageHeader_Bye.toField       := p_SessionInProgress183.msgHeader.toField;                  

      v_MessageHeader_Bye.fromField     := cs_From(v_InviteToAddrSpec, tsc_IMS_InviteToTag);       // !!!! CHECK !!!!
; 
      // scratch: cs_From(v_InviteToAddrSpec, f_MessageHeader_GetToTag(p_MessageHeader_PrevReliableResponse))

    } else { // MT call

      v_MessageHeader_Bye.toField       := p_SessionInProgress183.msgHeader.toField;  

      v_MessageHeader_Bye.fromField     := v_MessageHeader_Invite.fromField;        

    }

    return v_MessageHeader_Bye;

  }
Change 4 – f_IMS_BuildSDP_AnnexC11_Step1or7
	Function name
	 f_IMS_BuildSDP_AnnexC11_Step1or7

	Reason for change
	Steps 1 and 7 have one more difference than the sendrecv attribute: when executing step 7, resources at the remote end are reserved and need to be signalled.

	Summary of change
	Restructured the code to have a full listing on what is in step 1 and what is in step 7. For now, the differences in step 7 (as compared to step 1) are in the cs_SDP_Attribute_curr_quos and the already existing change to v_SDP_Attribute_List. 

	TTCN module
	 Common/IMS/IMS_SDP_Messages

	MCC160 Comment
	


Before change

  function f_IMS_BuildSDP_AnnexC11_Step1or7(charstring p_Step,

                                            template (value) SDP_attribute p_SDP_Attribute_curr_quos_remote := cs_SDP_Attribute_curr_quos(c_remote))

    runs on IMS_PTC return template (value) SDP_Message

  { /* Build SDP record to be sent to the UE at step 1 according to 34.229-1 Annex C.11 */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrOrigin := f_IMS_PTC_NW_Address_GetAddrStr();

    var charstring v_FmtAudio := tsc_SDP_FmtAudio;

    var template (value) SDP_media_desc v_SDP_Media_Desc;

    var template (omit) SDP_attribute_list v_SDP_Attribute_List := {
      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

      cs_SDP_Attribute_fmtp(v_FmtAudio, tsc_AMR_Fmtp_DefaultParameters),

      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime,

      // precondition attributes:

      cs_SDP_Attribute_curr_quos(c_local),

      p_SDP_Attribute_curr_quos_remote,

      cs_SDP_Attribute_des_quos(c_mandatory, c_local),

      cs_SDP_Attribute_des_quos(c_optional, c_remote)

    };

    if (p_Step == "step 7") {

      v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_sendrecv);

    }

    v_SDP_Media_Desc := cs_SDP_Media_Desc(cs_SDP_Media_Audio(v_FmtAudio),

                                          cs_SDP_Bandwidth_List_Media(30, 0, 0),  // @sic R5-130751 sic@

                                          v_SDP_Attribute_List);

    return cs_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrOrigin, tsc_SDP_Sessionname, v_SDP_Media_Desc);

  }
After change

  function f_IMS_BuildSDP_AnnexC11_Step1or7(charstring p_Step,

                                            template (value) SDP_attribute p_SDP_Attribute_curr_quos_remote := cs_SDP_Attribute_curr_quos(c_remote))

    runs on IMS_PTC return template (value) SDP_Message

  { /* Build SDP record to be sent to the UE at step 1 according to 34.229-1 Annex C.11 */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrOrigin := f_IMS_PTC_NW_Address_GetAddrStr();

    var charstring v_FmtAudio := tsc_SDP_FmtAudio;

    var template (value) SDP_media_desc v_SDP_Media_Desc;

    var template (omit) SDP_attribute_list v_SDP_Attribute_List; 

    if (p_Step == "step 1") {                                    

     v_SDP_Attribute_List := {         

      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

      cs_SDP_Attribute_fmtp(v_FmtAudio, tsc_AMR_Fmtp_DefaultParameters),

      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime,

      // precondition attributes:

      cs_SDP_Attribute_curr_quos(c_local),

      p_SDP_Attribute_curr_quos_remote,

      cs_SDP_Attribute_des_quos(c_mandatory, c_local),

      cs_SDP_Attribute_des_quos(c_optional, c_remote)

    };

    }                                       

    if (p_Step == "step 7") {               
      v_SDP_Attribute_List := {

         // media attributes:

         cs_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

         cs_SDP_Attribute_fmtp(v_FmtAudio, tsc_AMR_Fmtp_DefaultParameters),

         cs_SDP_Attribute_ptime,

         cs_SDP_Attribute_maxptime,

         // precondition attributes:

         cs_SDP_Attribute_curr_quos(c_local, c_sendrecv),        
         p_SDP_Attribute_curr_quos_remote,

         cs_SDP_Attribute_des_quos(c_mandatory, c_local),

         cs_SDP_Attribute_des_quos(c_optional, c_remote)

       };  

      v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_sendrecv);

    }

    v_SDP_Media_Desc := cs_SDP_Media_Desc(cs_SDP_Media_Audio(v_FmtAudio),

                                          cs_SDP_Bandwidth_List_Media(30, 0, 0),  // @sic R5-130751 sic@

                                          v_SDP_Attribute_List);

    return cs_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrOrigin, tsc_SDP_Sessionname, v_SDP_Media_Desc);

  }
Change 5 – f_IMS_BuildSDP_AnnexC11_Step4or8
	Function name
	 f_IMS_BuildSDP_AnnexC11_Step4or8

	Reason for change
	UEs send more than a format type number in 183 Session Progress (step 4) and 200 OK for Update (step 8). 34.229-1 accomodates for this by allowing any format.

	Summary of change
	List wildcard to allow any format.

	TTCN module
	Common/IMS/IMS_SDP_Messages

	MCC160 Comment
	


Before change

function f_IMS_BuildSDP_AnnexC11_Step4or8(charstring p_Step) runs on IMS_PTC return template (present) SDP_Message

  { /* to receive SDP answer sent by the UE in C.11 step 8 (200 OK sent by the SS) or 2nd SDP sent by the UE in C.21 step 5/7 */

    /* @sic R5-123077 change 3: 'inactive' removed from 'Attributes for media' for step 4 sic@ */

    var charstring v_FmtAudio := tsc_SDP_FmtAudio;

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List;

    select (p_Step) {

      case ("step 4") {

        v_SDP_Attribute_List := {

          cr_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

          cr_SDP_Attribute_fmtp(v_FmtAudio),
          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, (c_none, c_sendrecv)),     // @sic R5s120530 change 3.15 sic@

          cr_SDP_Attribute_curr_quos(c_remote, c_none),

          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case ("step 8") {

        v_SDP_Attribute_List := {

          // media attributes:

          cr_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

          cr_SDP_Attribute_fmtp(v_FmtAudio),
          cr_SDP_Attribute_sendrecv,

          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, c_sendrecv),

          cr_SDP_Attribute_curr_quos(c_remote, c_sendrecv),

          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(v_FmtAudio),

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
After change

  function f_IMS_BuildSDP_AnnexC11_Step4or8(charstring p_Step) runs on IMS_PTC return template (present) SDP_Message

  { /* to receive SDP answer sent by the UE in C.11 step 8 (200 OK sent by the SS) or 2nd SDP sent by the UE in C.21 step 5/7 */

    /* @sic R5-123077 change 3: 'inactive' removed from 'Attributes for media' for step 4 sic@ */

    var charstring v_FmtAudio := tsc_SDP_FmtAudio;

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List;

    select (p_Step) {

      case ("step 4") {

        v_SDP_Attribute_List := {

          cr_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

          cr_SDP_Attribute_fmtp(v_FmtAudio, {"*"}),                               
          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, (c_none, c_sendrecv)),     // @sic R5s120530 change 3.15 sic@

          cr_SDP_Attribute_curr_quos(c_remote, c_none),

          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case ("step 8") {

        v_SDP_Attribute_List := {

          // media attributes:

          cr_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

          cr_SDP_Attribute_fmtp(v_FmtAudio, {"*"}),                         

          cr_SDP_Attribute_sendrecv,       

          // precondition attributes:

          cr_SDP_Attribute_curr_quos(c_local, c_sendrecv),

          cr_SDP_Attribute_curr_quos(c_remote, c_sendrecv), 
          cr_SDP_Attribute_des_quos(c_mandatory, c_local),

          cr_SDP_Attribute_des_quos(c_mandatory, c_remote),

          *

        };

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(v_FmtAudio),

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
Change 6 – f_IMS_InviteResponse_183_MessageHeaderRX
	Function name
	 f_IMS_InviteResponse_183_MessageHeaderRX

	Reason for change
	1. Record-Route header in a response should be simply copied from request, see RFC 3261
2. Rseq is a number chosen by the responding party, here the UE. It cannot be prescribed by the server side

	Summary of change
	1. Copy Record-Route header from Invite request. Note that the Record-Route entries in 34.229-1, A.2.9 condition A1 and A.2.3 condition A2 already correspond but that it is still to be investigated if all handling of Record-Route is fine in 34.229-1
2. Allow any value for rSeq.

	TTCN module
	 Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	


Before change

  function f_IMS_InviteResponse_183_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                                    boolean p_IsGiba := false) runs on IMS_PTC return template (present) MessageHeader

  { // 183 Session Progress for INVITE (see TS 34.229, A.2.3)

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var integer v_Port_us := f_IMS_PTC_GetPort_us(not p_IsGiba);

    var template (present) charstring v_ToTag := ?;

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.recordRoute     := fl_IMS_RecordRoute_A23_A31_RX(p_IsGiba, f_IMS_PTC_ImsInfo_GetSigCompParam());
    v_MessageHeader_Response.require         := cr_Require("100rel");

    v_MessageHeader_Response.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Response.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, v_ToTag);

    v_MessageHeader_Response.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us), cr_FeatureParamInContactParams(?));    // host address needs to be checked after receiving

    v_MessageHeader_Response.rSeq            := cr_RSeq(tsc_IMS_RSeqNumFor183);
    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    v_MessageHeader_Response.contentType     := cr_ContentTypeDef("application/sdp");

    return v_MessageHeader_Response;

  }
After change

  function f_IMS_InviteResponse_183_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                                    boolean p_IsGiba := false) runs on IMS_PTC return template (present) MessageHeader

  { // 183 Session Progress for INVITE (see TS 34.229, A.2.3)

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var integer v_Port_us := f_IMS_PTC_GetPort_us(not p_IsGiba);

    var template (present) charstring v_ToTag := ?;

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.recordRoute     := v_MessageHeader_Invite.recordRoute;    v_MessageHeader_Response.require         := cr_Require("100rel");

    v_MessageHeader_Response.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Response.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, v_ToTag);

    v_MessageHeader_Response.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us), cr_FeatureParamInContactParams(?));    // host address needs to be checked after receiving

    v_MessageHeader_Response.rSeq            := ?; 
    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    v_MessageHeader_Response.contentType     := cr_ContentTypeDef("application/sdp");

    return v_MessageHeader_Response;

  }
Change 7 – f_IMS_MTCallSetup_AnnexC11_Steps5_13
	Function name
	f_IMS_MTCallSetup_AnnexC11_Steps5_13

	Reason for change
	UPDATE and ACK need to carry the to-tag that was established for this dialog. This to-tag was first seen in 183 Session Progress 

	Summary of change
	Feed the previous reliable response into subfunctions to take the to-tag from.
Note: other users of subfunction f_IMS_AckRequest_MessageHeaderTX need this fix as well, but the fixes will differ, e.g., the to-tag may need to be taken from a response other than 183 Session Progress. For the time being, a new function f_IMS_AckRequest_MessageHeaderTX2 is being proposed here (see next change for the new function). If this approach is acceptable, the submitter will bring a follow-up CR to fix the other places regarding the To-tag for ACK requests. For TC 12.13 though this local fix is sufficient.

	TTCN module
	Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	


Before change

function f_IMS_MTCallSetup_AnnexC11_Steps5_13(template (value) INVITE_Request p_InviteRequest,

                                                IMS_Response_Type p_SessionInProgress183) runs on IMS_PTC

  {

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Prack;

    var template (value) MessageHeader v_MessageHeader_Update;

    var template (value) MessageHeader v_MessageHeader_Ack;

    var MessageHeader v_MessageHeader_PrevReliableResponse := p_SessionInProgress183.msgHeader;

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var charstring v_Direction;

    var boolean v_Ringing := false;

    var template (value)   SDP_Message v_Tx_SDP;

    var template (present) SDP_Message v_Rx_SDP;

    timer t_Wait := 5.0;

    // Step 5. Send PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderTX(p_InviteRequest, v_MessageHeader_PrevReliableResponse);

    IMS_Client.send(cas_IMS_Prack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack)));

    // Step 6. Receive 200 OK

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Prack))));

    // Step 7. Send UPDATE

    // get curr:qos remote for SDP message in step 7

    if (isvalue(p_SessionInProgress183.messageBody.sdpMessageBody.media_list[0]) and

        f_SDP_AttributeList_CheckAttribute(p_SessionInProgress183.messageBody.sdpMessageBody.media_list[0].attributes, cr_SDP_Attribute_curr_quos(c_local, c_none))) {  // see C.11 step 7 Note 1

      v_Direction := c_none;

    } else {

      v_Direction := c_sendrecv;

    }

    v_Tx_SDP := f_IMS_BuildSDP_AnnexC11_Step1or7("step 7", cs_SDP_Attribute_curr_quos(c_remote, v_Direction));

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(p_InviteRequest);
    IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_Protocol), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(v_Tx_SDP))));

    // Step 8. Receive 200 OK. The UE acknowledges the UPDATE with 200 OK and includes SDP answer to acknowledge its current precondition status.

    v_Rx_SDP := f_IMS_BuildSDP_AnnexC11_Step4or8("step 8"); // @sic R5s120530 change 3.16 sic@

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200,

                                                                                    f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update),

                                                                                    cr_MessageBody_SDP(v_Rx_SDP))))

      -> value v_IMS_DATA_RSP;

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections (v_IMS_DATA_RSP.Response.messageBody.sdpMessageBody);

    t_Wait.start; // @sic R5-125772 sic@ Have introduced the timer to accommodate optional messages that may not arrive

    alt {

      // Step 9. Receive optional 180 Ringing

      [not v_Ringing] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderRX(p_InviteRequest))))

        -> value v_IMS_DATA_RSP

        {

          t_Wait.stop;

          v_Ringing :=  true;

          v_MessageHeader_PrevReliableResponse := v_IMS_DATA_RSP.Response.msgHeader;

          f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader_PrevReliableResponse);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

          if (f_MessageHeader_CheckRequire(v_MessageHeader_PrevReliableResponse, "100rel")) {

            // Step 10. Send PRACK

            v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderTX(p_InviteRequest, v_MessageHeader_PrevReliableResponse);

            IMS_Client.send(cas_IMS_Prack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack)));

            // Step 11. The UE acknowledges the PRACK with 200 OK

            IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Prack))));

          }

          // Step 11A - Make UE accept the speech AMR offer

          f_UT_IMS_AcceptCall(MMI);

          repeat;

        }

      [] t_Wait.timeout

        {

          // Step 11A - Make UE accept the speech AMR offer

          f_UT_IMS_AcceptCall(MMI);

          repeat;

        }

      // Step 12. Receive 200 OK. The UE responds to INVITE with a 200 OK final response after the user answers the call.

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_InviteResponse_200_MessageHeaderRX(v_MessageHeader_Invite)))) -> value v_IMS_DATA_RSP

        {

          f_IMS_MessageHeader_CheckContactAddr(v_IMS_DATA_RSP.Response.msgHeader);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

        }

    } // alt

    // Step 13. Send ACK. The SS acknowledges the receipt of 200 OK for INVITE.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(p_InviteRequest);
    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

   // MT Call established

  }
After change

  function f_IMS_MTCallSetup_AnnexC11_Steps5_13(template (value) INVITE_Request p_InviteRequest,

                                                IMS_Response_Type p_SessionInProgress183) runs on IMS_PTC

  {

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Prack;

    var template (value) MessageHeader v_MessageHeader_Update;

    var template (value) MessageHeader v_MessageHeader_Ack;

    var MessageHeader v_MessageHeader_PrevReliableResponse := p_SessionInProgress183.msgHeader;

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var charstring v_Direction;

    var boolean v_Ringing := false;

    var template (value)   SDP_Message v_Tx_SDP;

    var template (present) SDP_Message v_Rx_SDP;

    timer t_Wait := 5.0;

    // Step 5. Send PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderTX(p_InviteRequest, v_MessageHeader_PrevReliableResponse);

    IMS_Client.send(cas_IMS_Prack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack)));

    // Step 6. Receive 200 OK

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Prack))));

    // Step 7. Send UPDATE

    // get curr:qos remote for SDP message in step 7

    if (isvalue(p_SessionInProgress183.messageBody.sdpMessageBody.media_list[0]) and

        f_SDP_AttributeList_CheckAttribute(p_SessionInProgress183.messageBody.sdpMessageBody.media_list[0].attributes, cr_SDP_Attribute_curr_quos(c_local, c_none))) {  // see C.11 step 7 Note 1

      v_Direction := c_none;

    } else {

      v_Direction := c_sendrecv;

    }

    v_Tx_SDP := f_IMS_BuildSDP_AnnexC11_Step1or7("step 7", cs_SDP_Attribute_curr_quos(c_remote, v_Direction));

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(p_InviteRequest, v_MessageHeader_PrevReliableResponse); 

    IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_Protocol), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(v_Tx_SDP))));

    // Step 8. Receive 200 OK. The UE acknowledges the UPDATE with 200 OK and includes SDP answer to acknowledge its current precondition status.

    v_Rx_SDP := f_IMS_BuildSDP_AnnexC11_Step4or8("step 8"); // @sic R5s120530 change 3.16 sic@

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200,

                                                                                    f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update),

                                                                                    cr_MessageBody_SDP(v_Rx_SDP))))

      -> value v_IMS_DATA_RSP;

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections (v_IMS_DATA_RSP.Response.messageBody.sdpMessageBody);

    t_Wait.start; // @sic R5-125772 sic@ Have introduced the timer to accommodate optional messages that may not arrive

    alt {

      // Step 9. Receive optional 180 Ringing

      [not v_Ringing] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderRX(p_InviteRequest))))

        -> value v_IMS_DATA_RSP

        {

          t_Wait.stop;

          v_Ringing :=  true;

          v_MessageHeader_PrevReliableResponse := v_IMS_DATA_RSP.Response.msgHeader;

          f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader_PrevReliableResponse);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

          if (f_MessageHeader_CheckRequire(v_MessageHeader_PrevReliableResponse, "100rel")) {

            // Step 10. Send PRACK

            v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderTX(p_InviteRequest, v_MessageHeader_PrevReliableResponse);

            IMS_Client.send(cas_IMS_Prack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack)));

            // Step 11. The UE acknowledges the PRACK with 200 OK

            IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Prack))));

          }

          // Step 11A - Make UE accept the speech AMR offer

          f_UT_IMS_AcceptCall(MMI);

          repeat;

        }

      [] t_Wait.timeout

        {

          // Step 11A - Make UE accept the speech AMR offer

          f_UT_IMS_AcceptCall(MMI);

          repeat;

        }

      // Step 12. Receive 200 OK. The UE responds to INVITE with a 200 OK final response after the user answers the call.

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_InviteResponse_200_MessageHeaderRX(v_MessageHeader_Invite)))) -> value v_IMS_DATA_RSP

        {

          f_IMS_MessageHeader_CheckContactAddr(v_IMS_DATA_RSP.Response.msgHeader);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

        }

    } // alt

    // Step 13. Send ACK. The SS acknowledges the receipt of 200 OK for INVITE.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX2(p_InviteRequest, v_MessageHeader_PrevReliableResponse);
    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

   // MT Call established

  }
Change 8 – f_IMS_AckRequest_MessageHeaderTX2
	Function name
	 f_IMS_AckRequest_MessageHeaderTX2

	Reason for change
	1. See Change 6 regarding the to-tag
2. ACK needs its own branch-ID, see condition A3 in 34.229-1, A.2.7, via-branch, condition A3

	Summary of change
	1. Take the to-tag from the previous reliable response
2. Build via from scratch

	TTCN module
	 Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	


Before change

  function f_IMS_AckRequest_MessageHeaderTX(template (value) INVITE_Request p_InviteRequest) runs on IMS_PTC return template (value) MessageHeader

  { /* ACK (see TS 34.229, A.2.7) */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Ack := cs_MessageHeader_Dummy;

    var template (value) Via v_Via := fl_Via_AddSigCompParam(v_MessageHeader_Invite.via);
    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);    // for ACK the CSeq value of the corresponding INVATE has to be used

    var charstring v_ToTag := f_IMS_GenerateTag("ACK");   // !!!! might be wrong, prose clarification needed !!!!
    v_MessageHeader_Ack.via             := v_Via;

    v_MessageHeader_Ack.toField         := cs_ToWithTag(f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.toField.addressField), v_ToTag);

    v_MessageHeader_Ack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.cSeq            := cs_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cs_MaxForwardsDef;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    return v_MessageHeader_Ack;

  }
After change

function f_IMS_AckRequest_MessageHeaderTX2(template (value) INVITE_Request p_InviteRequest,

                                             MessageHeader p_MessageHeader_PrevReliableResponse 

                                            ) runs on IMS_PTC return template (value) MessageHeader

  { /* ACK (see TS 34.229, A.2.7) */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Ack := cs_MessageHeader_Dummy;

    var template (value) Via v_Via;  //assign v_Via  later down in the function
    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);    // for ACK the CSeq value of the corresponding INVATE has to be used

    var charstring v_ToTag := f_MessageHeader_GetToTag(p_MessageHeader_PrevReliableResponse); 

    var charstring v_Host := f_IMS_PTC_Pcscf_Get();

    var charstring v_TransportProtocol := "UDP";

    var integer v_CallerPort := 6543;

    var template (value) ViaBody_List v_ViaBodyList;

    var charstring v_CallerAddr := tsc_IMS_CallerDomain;

    var boolean p_IsGiba := false;

    var integer v_Port_ps := f_IMS_PTC_GetPort_ps(not p_IsGiba);

    var template (omit) GenericParam v_SigCompParam := f_IMS_PTC_ImsInfo_GetSigCompParam();  // omit when sigcomp is not started

    v_ViaBodyList := {

      cs_ViaBody(v_TransportProtocol, cs_HostPort(v_Host, v_Port_ps),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch(), v_SigCompParam)),

      cs_ViaBody(v_TransportProtocol, cs_HostPort("scscf1.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol, cs_HostPort("scscf2.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol, cs_HostPort("pcscf2.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol, cs_HostPort(v_CallerAddr, v_CallerPort), f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch()))

    };  

    v_Via := cs_Via(v_ViaBodyList);  
    v_MessageHeader_Ack.via             := v_Via;

    v_MessageHeader_Ack.toField         := cs_ToWithTag(f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.toField.addressField), v_ToTag);

    v_MessageHeader_Ack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.cSeq            := cs_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cs_MaxForwardsDef;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    return v_MessageHeader_Ack;

  }
Change 9 – f_IMS_InviteResponse_180_MessageHeaderRX
	Function name
	 f_IMS_InviteResponse_180_MessageHeaderRX

	Reason for change
	Record-Route is still to be fixed in prose. 

	Summary of change
	In the meantime, for this call, take the entries from the else clause in fl_IMS_RecordRoute_A23_A31_RX

	TTCN module
	 Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	


Before change

  function f_IMS_InviteResponse_180_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                                    boolean p_IsGiba := false,

                                                    boolean p_IsEmergency := false) runs on IMS_PTC return template (present) MessageHeader

  { // 180 Ringing for INVITE (see TS 34.229, A.2.6)

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var integer v_Port_us := f_IMS_PTC_GetPort_us(not p_IsGiba);

    var template (present) charstring v_ToTag := ?;

  v_MessageHeader_Response.recordRoute     := fl_IMS_RecordRoute_A23_A31_RX(p_IsGiba);  //UE2_TC#12_13, ETSI code

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Response.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, v_ToTag);

    v_MessageHeader_Response.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us), cr_FeatureParamInContactParams(?));    // host address needs to be checked after receiving

    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    v_MessageHeader_Response.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);    // NOTE: acc. A.2.6 there is no condition regarding GIBA

    if (p_IsEmergency) {    // Condition A4

      v_MessageHeader_Response.pAssertedID      := cr_PAssertedId(f_SIP_BuildSipUri_lr_RX(tsc_IMS_Emergency_TelUri));

    }

    return v_MessageHeader_Response;

  }
After change

  function f_IMS_InviteResponse_180_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                                    boolean p_IsGiba := false,

                                                    boolean p_IsEmergency := false) runs on IMS_PTC return template (present) MessageHeader

  { // 180 Ringing for INVITE (see TS 34.229, A.2.6)

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var integer v_Port_us := f_IMS_PTC_GetPort_us(not p_IsGiba);

    var template (present) charstring v_ToTag := ?;

    v_MessageHeader_Response.recordRoute     := fl_IMS_RecordRoute_A23_A31_RX(p_IsGiba, - , true); 
    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Response.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, v_ToTag);

    v_MessageHeader_Response.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us), cr_FeatureParamInContactParams(?));    // host address needs to be checked after receiving

    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    v_MessageHeader_Response.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);    // NOTE: acc. A.2.6 there is no condition regarding GIBA

    if (p_IsEmergency) {    // Condition A4

      v_MessageHeader_Response.pAssertedID      := cr_PAssertedId(f_SIP_BuildSipUri_lr_RX(tsc_IMS_Emergency_TelUri));

    }

    return v_MessageHeader_Response;

  }
Change 10 – f_IMS_UpdateRequest_MessageHeaderTX
	Function name
	f_IMS_UpdateRequest_MessageHeaderTX

	Reason for change
	Update request must use the to-tag first seen in 183.

	Summary of change
	Feed the 183 response into the function for Update and use it for the to-tag. Note: it is still to be investigated if this is appropriate for other scenarios where Update is used. The fix proposed here works as long as there is a response of the UE preceding the Update (it is not necessary this response to be 183).

	TTCN module
	Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	


Before change

function f_IMS_UpdateRequest_MessageHeaderTX(template (value) INVITE_Request p_InviteRequest) runs on IMS_PTC return template (value) MessageHeader //UE2_TC#12_13, ETSI code

  { /* UPDATE (see TS 34.229, A.2.5) */

    /* p_CSeq   ..   "value of CSeq sent by the endpoint within its previous request in the same dialog" */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Update := cs_MessageHeader_Dummy;

    var template (value) Via v_Via := fl_Via_AddSigCompParam(v_MessageHeader_Invite.via);

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMT);

    var charstring v_ToTag := f_IMS_GenerateTag("UPDATE");   // !!!! might be wrong, prose clarification needed !!!!
    v_MessageHeader_Update.via             := v_Via;

    v_MessageHeader_Update.toField         := cs_ToWithTag(f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.toField.addressField), v_ToTag);

    v_MessageHeader_Update.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Update.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Update.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Update.cSeq            := cs_CseqDef(v_CSeqValue, "UPDATE");

    v_MessageHeader_Update.maxForwards     := cs_MaxForwardsDef;

    v_MessageHeader_Update.contentType     := cs_ContentTypeDef("application/sdp");

    return v_MessageHeader_Update;

  }
After change

  function f_IMS_UpdateRequest_MessageHeaderTX(template (value) INVITE_Request p_InviteRequest,

                                               MessageHeader p_MessageHeader_PrevReliableResponse
                                              ) runs on IMS_PTC return template (value) MessageHeader

  { /* UPDATE (see TS 34.229, A.2.5) */

    /* p_CSeq   ..   "value of CSeq sent by the endpoint within its previous request in the same dialog" */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Update := cs_MessageHeader_Dummy;

    var template (value) Via v_Via := fl_Via_AddSigCompParam(v_MessageHeader_Invite.via);

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMT);

    var charstring v_ToTag := f_MessageHeader_GetToTag(p_MessageHeader_PrevReliableResponse); 
    v_MessageHeader_Update.via             := v_Via;

    v_MessageHeader_Update.toField         := cs_ToWithTag(f_Addr_Union_GetSipUrl(v_MessageHeader_Invite.toField.addressField), v_ToTag);

    v_MessageHeader_Update.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Update.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Update.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Update.cSeq            := cs_CseqDef(v_CSeqValue, "UPDATE");

    v_MessageHeader_Update.maxForwards     := cs_MaxForwardsDef;

    v_MessageHeader_Update.contentType     := cs_ContentTypeDef("application/sdp");

    return v_MessageHeader_Update;

  }
5. Execution Log Files

Intel XMM7160 UE passed this test case on the Rohde & Schwarz Multi-RAT Protocol Tester. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_12_13_Intel_XMM7160.log

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
6. References

[1] R5s130511: Supporting information for re-verification of IMS test case 12.13 over LTE with 34.229-3 test model - This archive comprises html format execution log file

